
Introduction

The binomial option pricing model is a powerful numerical method to value options. Implementing a small binomial tree in a spreadsheet is a matter of minutes. However, achieving accuracy requires a large number of time steps (or periods). A minimum of 30 steps is often mentioned.

As an illustration, figure 1 shows the percentage error when valuing a six-month at the money American put option for different number of time steps1. Note that the number of steps necessary to achieve a valuation error below 0.5% is greater than 50.

Figure 1

Valuation error for an American put option as a function of the number of steps.

As the number of time periods in the tree increases, the number of nodes increases with the square of the number of time steps. Denoting as n the number of time steps, the number of nodes is (n²+n)/2. For instance, a binomial tree with 40 time steps would involve 820 nodes. Creating an Excel spreadsheet with such a large number of node is feasible but cumbersome. In addition, any change in the number of time steps would require setting up a new worksheet.


The shortcoming of using a spreadsheet is even more pronounced for other numerical methods2 such as the explicit method (a discrete approximation of the Black-Merton-Scholes partial differential equation) or Monte Carlo analysis. Advanced numerical methods are simply impossible to implement in a spreadsheet. The implicit method, for instance, requires solving a set of simultaneous equations at each step.


A more powerful and more flexible computation method is called for.


Using Visual Basic (VB), the programming language included in Excel, is one step toward this objective. It’s available in any Excel program. It would be a pity not to take advantage of it.


Learning a programming language takes a lot of time. But the investment required to master VB can be limited by restricting the notions to what is strictly necessary for the task at hand: option valuation. This is the objective of this note3.
Starting a VB program

Programs in Excel are referred to as functions and macros. They are stored in special Excel sheets called modules. A function calculates one single output. A macro can produce any number of results.


To write a function or a macro, first you have to access the Visual Basic editor. This can be done directly with Alt+F11 or through the menu Tools(Macro(Visual Basic Editor.


To leave the Visual Basic Editor, use Alt+Q or use the menu File(Close and Return to Microsoft Excel.


To create a module, use the Insert(Module menu in VB. Later on, you can access a module through the View(Project Explorer (shortcut Alt+R) in VB. A list appears on the right hand side of the screen. The existing modules are in this list. Double click on them to open the module sheet.

Functions

VB functions behave exactly like built-in functions found in Excel such as Sum or StDev. Inputs are provided by the arguments. The output will appear in cells where the function is used. The value is calculated whenever the workbook recalculates.


The format of a function is:

Function MyFunction(Arguments)

Code lines

End Function

It always starts with Function followed by the name that you choose for the function (for instance, My Function). The arguments are the data that you need to execute the function. Then come the code lines (the instructions that you want the function to execute). It always ends with End Function.

Example 1: Function Addone

Let’s illustrate this with a very simple example. We want to create a function that simply adds one to a cell value. We will name this function Addone.

The first step is to create a module as explained above. Then type the code in the module sheet. The first line tells that this is a function, gives the name you have chosen and the list of arguments. Note that Excel provides help as you type. For instance, after you have type the first line of the program (Function Addone(n)), Excel will automatically generate the last line (End Function).


The function has one single argument, the initial number (call it n). The second line starts with a single quotation mark. This is a comment. This line is not required to execute the program. Excel will ignore everything to the right of the quotation mark. Including comments in a program is a way to document the program. It greatly helps to understand what the program does. Note that comments appear in a different color than the code line on your screen.


The next line is the single code line. It calculates the value of the function as n+1.

Function Addone(n)

‘Add one to n

Addone = n+1

End Function
Using this function is a spreadsheet would look like this:


	
	A
	B

	1
	5
	6

	2
	
	


We will now go through a number of examples that illustrate various aspects of functions. The examples have been chosen with the final objective in mind of having a function to price any option with a binomial model. Each of these examples will constitute one building block of the full program. 

In order to calculate the value of the option, we will use the following data (next to each variable, we indicate the name of the cell)

Type of option: Call or Put (CallPut)

American or European (AmEur)

Stock Price (StockPrice)

Volatility (Volatility)

Interest Rate (IntRate)

Strike price (Strike)

Maturity (Maturity)

Number of steps (NoSteps)

Example 2: Function IntVal


As a second example, suppose that we want to have a function that calculates the intrinsic value of an option. Recall that the intrinsic value is:


Max(0, S – X) for a call option


Max (0, X – S) for a put option

where S is the stock price and X is the strike price. To have one single formula for both type of options, define a binary variable b that takes two values:

b = + 1 for a call option

b = - 1 for a put option.

Using the binary variable is a useful trick to use to avoid having to write separate programs for calls and puts.

The intrinsic value of the option is:

Max(0, b(S – X))

Indeed:

For a call b = +1: Max(0, b(S – X)) = Max(0, S – X)

For a put b = -1: Max(0, b(S – X)) = Max(0, X – S)

The VB code for this function is:

Function IntVal(s,x,b)

‘Calculate the intrinsic value

IntVal = b*(s-x)

If IntVal < 0 Then


IntVal = 0

End If

End Function

This function has three arguments: s is the stock price, x the exercise price and b the binary variable providing the information on the type of option.
Note the If condition used in this program. After calculating the intrinsic value as b(S-X), the program then checks whether this value is negative. If so, it sets the intrinsic value equal to 0.

The following table illustrates the use of the function. It calculates the intrinsic value of a call option when the stock price is 100 and the exercise price is 80. Data are in cells B1, B2 and B3. Cell B4 invokes the function as =IntVal(B2;B3;B1)

	
	A
	B

	1
	b
	1

	2
	S
	100

	3
	X
	80

	4
	V
	20


Example 3: AssetPrice
AssetPrice is a function that calculates the stock price at a node in a binomial tree. Let S0 be the initial price, n be the number of periods, nd be the number of downs and u be the up factor. The down factor is d = 1/u.

The formula is: S0 un-nddnd
The VB code is:

Function AssetPrice(s0,u,n,nd)

‘Calculates asset price at period n after nd downs


d = 1/u

‘d is the down factor


nu = n – nd
‘nu is the number of ups


AssetPrice = s0*u^nu*d^nd

End Function

The code is straight forward. Note that a comment can be written on a line. It should start with a single quotation mark. Note to the syntax to calculate the power of a number.

Example 4: BinCP

Function BinCP calculates the value of the binary function b defined in example 3. It uses as argument OptType, a set of characters (Call or Put) called a string. The type of option is identified by checking the first character of the string.

The VB code for this function is:

Function BinCP(OptType as String)

‘Check whether the option is a call or a put

‘Check first letter to determine the type of option

If Left(OptType, 1) = “C” or Left(OptType, 1) = “c” Then BinCP = 1

If Left(OptType, 1) = “P” or Left(OptType, 1) = “p” Then BinCP = - 1

End Function

Note that the function returns 0 (zero) for any other name.
Example 5: BinAE

Function BinAE is similar to BinCP. It determines whether the option is European or American. The value of the option is 1 if the option is American and 0 if the option is European. The function uses as argument AmEur, a string (American or European).

The VB code for this function is:

Function BinAE(AmEur as String)

‘Check whether the option is American or European

‘Check first letter to determine the type of option

If Left(AmEur, 1) = “E” 
or Left(AmEur, 1) = “e” Then BinAE = 0

If Left(AmEur, 1) = “A” 
or Left(AmEur, 1) = “a” Then BinAE = 1

End Function

Example 6: OptMat01

OptMat01 is a program that calculates one single value of an option at maturity. The arguments of the function are:

OptType:
the type of option (Call or Put)

StockPrice:
the initial stock price

Strike:

the strike price

Volatility:
the volatility of the stock

Maturity:
the maturity of the option (expressed in years)

NoSteps:
the number of time steps (at least 1)

Nd:

the number of downs

Two new features of VB are used in this function.

First, this code uses two VB internal function to calculate u (2d line in the code):

exp(x) to calculate ex
sqr(x) to calculate the square root

Second, the function uses two functions that where defined previously: AssetPrice to calculate the stock price at maturity and IntVal to calculate the intrinsic value.
The VB code is:

Function OptMat01(OptType As String, StockPrice, Strike, Volatility, Maturity, NoSteps, nd)

    Timestep = Maturity / NoSteps               'Calculate time step

    u = Exp(Volatility * Sqr(Timestep))         'Calculate up factor

    s = AssetPrice(StockPrice, u, NoSteps, nd)  'Calculate stock price at maturity

    b = BinCP(OptType)                          'Determine type of option

    OptMat01 = IntVal(s, Strike, b)             'Calculate value at maturity

End Function

Example 7: OptMat02
Let’s go one step further. Function OptMat01 calculates one single value for the option at maturiy. But to calculate the initial value of the option, we will need later to start from all possible values at maturity. We will store these values in a vector. Let’s denote that vector as V. The number of elements of this vector is NoSteps+1 (remember, NoSteps is the number of steps in the binomial tree). Element j (j= 0, 1, …, NoSteps) in this vector is V(j). The program should know what the size of the vector is. This is done by a declaration at the beginning of the code.

To store the values, we will use a loop. Loops are extremely powerful tools. A loop tells the program to repeat one or several instruction a number of times. The number of time is indicated by a counter. You can choose the name for this counter. Suppose j is the name for the counter and n is the number of time to repeat the instruction. The general format of a loop would be:

For j = 0 to n

Instructions

Next j

This code tells Excel to repeat the instruction for different values of j starting with j=0 and ending with j=n. Unless specified, the step for j is +1.

OptMat02 is based on OptMat01 with additional lines (in bold to facilitate the comparison).

The VB code is:

Function OptMat02(OptType As String, StockPrice, Strike, Volatility, Maturity, NoSteps, nd)

    ReDim V(0 To NoSteps)                       

'Specify size of vector V

    Timestep = Maturity / NoSteps               'Calculate time step

    u = Exp(Volatility * Sqr(Timestep))         'Calculate up factor

    b = BinCP(OptType)                          'Determine type of option

    For j = 0 To NoSteps                        

'Start loop

         s = AssetPrice(StockPrice, u, NoSteps, j)  'Calculate stock price at maturity

         V(j) = IntVal(s, Strike, b)            'Calculate option value

    Next j                                      
'End loop

    OptMat02 = V(nd)                            
'Return element asked in argument

End Function

Example 8: OptionValue

We will now create OptionValue, the option pricing function. It moves back through the tree to obtain the current value. This will require several changes in the function OptMat02.

First, we modify the list of arguments. We include a variable telling whether the option is European or American. As in example 3, this information is provided by a string (American or European). We also need to include the interest rate (with continuous compounding). On the other, the number of downs (nd) is no longer necessary.

Some additional lines are necessary to calculate the discount factor to use at each node and the risk neutral probability. We also use the BinAE function introduced in example 4.

Finally, we introduce an additional loop to move back through the tree. This is somewhat tricky. To understand the loop, consider a case with NoSteps = 3.

The following table illustrate the content of vector V (which at each period).

	Down
	Period 3
	Period 2
	Period 1
	Period 0

	0
	Vuuu
	Vuu
	Vu
	V

	1
	Vuud
	Vud
	Vd
	

	2
	Vudd
	Vdd
	
	

	3
	Vddd
	
	
	


We first calculate the different values of the option at maturity (period 3) as explained in the function OptMat(02). We then move back through the tree using a loop. This loop start calculating the values of the option at period 2 (1 period before the final maturity). It should then decrease the period number by one and repeat the calculation. It should go on until it reaches period 0.

For each period, several values for the option have to be calculated. This is done with a second loop embedded in the first loop. The counter for this second loop goes from 0 to n, where n is the current period number.

So, the structure of the code looks like this:

For n = NoPeriod – 1 to 0 Step -1

For j = 0 to n

Instructions

Next j

Next n

Let us now look at the instructions.

First we calculate the value of the option based on the value in adjacent notes. The code is:

V(j) = (p*V(j)+(1-p)*V(j+1))*DiscountFactor
This formula is simply the one-period binomial valuation formula with some additional subtleties. The values of V on the right-hand side have been calculated in the previous period. The equality sign tells the program to store the result of the calculation in V(j). V(j) is now the value for the current period. But V(j+1) is left intact; it’s still the value from the previous period. When j is incremented by on to j+1, the calculation is:

V(j+1) = (p*V(j+1)+(1-p)*V(j+2))*DiscountFactor

The new value of V(j+1) is then calculated.
The next step is to check whether to exercise or not the option if the option is American. To do this, we first calculate the intrinsic value of the option at current node use the IntVal function (remember that the intrinsic value is the value of the option if exercised immediately). We then compare the intrinsic value to the value given by the binomial valuation formula (this is the value of the option if not exercised). We capture the American nature of the option by multiplying the intrinsic value by a binary variable taking a value of 1 is the option is American and 0 if the option is European. 
We get the following code:

Function OptionValue(OptType As String, AmEur As String, StockPrice, Strike, Volatility, IntRate, Maturity, NoSteps)

    ReDim V(0 To NoSteps)

    'Calculate parameters used in binomial tree

    TimeStep = Maturity / NoSteps

    DiscountFactor = Exp(-IntRate * TimeStep)

    u = Exp(Volatility * Sqr(TimeStep))

    d = 1 / u

    p = (Exp(IntRate * TimeStep) - d) / (u - d)

    'Binary variables to identify the type of option

    b = BinCP(OptType)

    BAE = BinAE(AmEur)

    'Calculate option value at maturity

    For j = 0 To NoSteps

         s = AssetPrice(StockPrice, u, NoSteps, j)

         V(j) = IntVal(s, Strike, b)

    Next j

    'Move back down the tree

    For n = (NoSteps - 1) To 0 Step -1

        For j = 0 To n

            V(j) = (p * V(j) + (1 - p) * V(j + 1)) * DiscountFactor

            s = AssetPrice(StockPrice, u, n, j)

            IntValue = IntVal(s, Strike, b)

            If IntValue * BAE > V(j) Then

                V(j) = IntValue

            End If

        Next j

    Next n

    OptionValue = V(0)

End Function
Macros

An important limitation of a function is that the output is limited to one single value. For instance, in the previous section, we developed a program to calculate the value of an option. But we might wish to have other results. For instance, in addition to the value of the option, we might want to have also the “greeks” (delta, gamma, theta, rho). To do this, we need to use a VB program, a macro.


Writing a macro is similar to writing a function with four differences:
a. the general format is different

b. the inputs are not provided by arguments as for a function

c. the outputs are not automatically transferred to the worksheet

d. the program is not automatically executed when the worksheet recalculates.

Let’s examine each of these point separately. We will then illustrate this with a macro that calculates the value and the delta of an option.

a. General format of a macro.

The format of a macro is:

Sub MyMacro()


Code lines

End Sub

It always start with Sub followed by the name that you choose for the macro. Note that there are no argument between the parenthesis. It always ends with End Sub.
b. Getting the input from a worksheet
Here I assume that the data that you want to use in the macro is stored in a worksheet. The instruction to use to retrieve this data is:

Variable = Worksheet(“sheetname”).Range(“cellname”).Value

This instruction tells VB to set the value of Variable equal to the value in the cell named cellname located in worksheet sheetname.

c. Exporting the output to a worksheet.

The instruction to use is:

Worksheet(“sheetname”).Range(“cellname”).Formula = yourformula

This instruction tells VB to put in cell cellname located in worksheet sheetname the result of yourformula.

Note that in order to avoid having to specify the worksheet each time for the input and the output, you can first specify which is the worksheet that you want use with the instruction:

ActiveSheet.Name = “MySheet”

A short example:
Sub Example()

ActiveSheet.Name=”Sheet1”

x = Range(“B1”).Value

y = Range(“B2”).Value

Range(“B3”).Formula = x + y

End Sub

In this example, numbers are stored in cells A1 and B1 of Sheet1. The program takes these values and stores their sum in cell C1.

d. Running a macro

To execute the program, you need to access a dialog box with the list of existing macros. This dialog box is reached through Tools(Macro(Macros (or Alt+F8). Choose the macro that you want to run and click the Run button.

A convenient way to run a program (macro) is to have a button. To create a button, use the View(Toolbar(Forms. Click the Button button (second line left) and draw the button somewhere in the worksheet. Excel displays the Assign Macro dialog box. Select the name of the macro that you want to assign to the button. While the Forms toolbar is active, you can edit the text appearing on the button. Close the Forms toolbar before using it. To execute the program then simply click the button.

Here is the VB code.

Sub ValueOption()

    ActiveSheet.Name = "Sheet1"

    Dim AmEur As String

    Dim CP As String

    Dim V1(0 To 1)

    Dim V2(0 To 2)

    'Get data

        CP = Range("CallPut").Value

        AmEur = Range("AmEur").Value

        StockPrice = Range("StockPrice").Value

        Strike = Range("Strike").Value

        Volatility = Range("Volatility").Value

        IntRate = Range("IntRate").Value

        Maturity = Range("Maturity").Value

        NoSteps = Range("NoSteps").Value

        ReDim V(0 To NoSteps)

    'Calculate parameters used in binomial tree

        TimeStep = Maturity / NoSteps

        DiscountFactor = Exp(-IntRate * TimeStep)

        u = Exp(Volatility * Sqr(TimeStep))

        d = 1 / u

        p = (Exp(IntRate * TimeStep) - d) / (u - d)

    'Binary variables to identify the type of option

        b = BinCP(CP)

        BAE = BinAE(AmEur)

    'Calculate option value at maturity

        For j = 0 To NoSteps

             s = AssetPrice(StockPrice, u, NoSteps, j)

             V(j) = IntVal(s, Strike, b)

        Next j

    'Move back down the tree

        For n = (NoSteps - 1) To 0 Step -1

            For j = 0 To n

                V(j) = (p * V(j) + (1 - p) * V(j + 1)) * DiscountFactor

                s = AssetPrice(StockPrice, u, n, j)

                IntValue = IntVal(s, Strike, b)

                If IntValue * BAE > V(j) Then

                    V(j) = IntValue

                End If

            Next j

            If n = 2 Then 'Store values period 2

                V2(0) = V(0)

                V2(1) = V(1)

                V2(2) = V(2)

            End If

            If n = 1 Then 'Store value period 1

                V1(0) = V(0)

                V1(1) = V(1)

            End If

        Next n

    'Calculate Greek letters

        Delta = (V1(0) - V1(1)) / (StockPrice * (u - d))

        g1 = (V2(0) - V2(1)) / (StockPrice * u * u - StockPrice)

        g2 = (V2(1) - V2(2)) / (StockPrice - StockPrice * d * d)

        h = 0.5 * StockPrice * (u * u - d * d)

        Gamma = (g1 - g2) / h

        Theta = (V2(1) - V(0)) / (2 * TimeStep)

    'Display results in worksheet

        Range("Value").Formula = V(0)

        Range("Delta").Formula = Delta

        Range("Gamma").Formula = Gamma

        Range("Theta").Formula = Theta

End Sub

A large fraction of the code is identical to the function OptionValue that we analyzed previously. Note the following changes:

1. the lines at the beginning to get the relevant data from the worksheet;

2. values of the option in period 2 and 1 are stored in two vectors (V2 and V1) so as to have these values at hand to compute delta, gamma and theta;

3. the Greek letters are calculated at the end

4. the program stores four results (the option value, delta, gamma and theta) in the worksheet. 


Notes

1. The data used to generate this figure are:

Type of option: American Put

Stock Price = 100

Strike Price = 100

Maturity = 0.5 year

Interest rate = 10%

Volatility = 40%

The “exact” value is calculated using 1,000 time steps.

For those of you willing to check their calculations, here are some numerical results


Number of time steps

Option’s value



5



9.6374



10



9.0856



15



9.3740



20



9.1527


25



9.3098



30



9.1750



35



9.2848



40



9.1860



1,000



9.2176





2. Hull (2002) Chap.18 is a good introduction to numerical procedures.

3. Textbooks on derivatives are silent about how to implement models in Excel. To my knowledge, the only exception is Beninga (1997) which includes an introduction to VB as an appendix. Some VB codes are also available in Wilmott (1998).
4. I have chosen the binomial model to introduce VB because of its simplicity (see Farber (2002) for a brief introduction to the model). Also any student having be exposed to option in a finance course is familiar with the model. It may not be the most powerful numerical method however. Finite difference methods are probably more accurate to value non standard options. For instance, Wilmott (1998, chap. 46) writes: ”On a personal note, for solving problems in practice I would say that I use finite-difference methods about 75% of the time, Monte Carlo simulations 20%, and the rest would be explicit formulae”.

The formula in this cell is ADDONE(A1)








